```
> restart;
  with(plots):
  with(linalg):
  interface(showassumed=0);
  assume(r>0);
  assume(epsilon>0);
  _EnvExplicit:=true:
```

$$0$$

Problem 1

```
> L:=u->diff(u,x,x)+diff(u,y,y);
  N:=u->(r-1)*u-2*L(u)-L(L(u))-u^5;
```

$$L := u \to \frac{\partial^2}{\partial x^2} u + \frac{\partial^2}{\partial y^2} u$$

$$N := u \to (r-1)\, u - 2\, L(u) - L(L(u)) - u^5$$

(a)

Linear stability of the uniform state u=0:

```
> eq:=collect(combine(coeff(N(0+epsilon*alpha*cos(q*x)),epsilon,1),
  trig),cos(q*x));
```

$$eq := \left(-\alpha\, q^4 + 2\,\alpha\, q^2 + \alpha\, r - \alpha\right) \cos(q\, x)$$

```
> sigma=factor(coeff(coeff(eq,cos(q*x),1),alpha,1));
```

$$\sigma = -q^4 + 2\, q^2 + r - 1$$

> # **This is the usual expression $\boldsymbol{\sigma}$=r$-(1-\mathbf{q}^2)^2$**

Zero imaginary part, maximum at q=1, so type-Is.

(b)

Stripe state:

```
> eq:=collect(combine(N(a1*cos(x))),{cos(x)});
  cf:=factor(coeff(eq,cos(x),1));
  sl:=solve({cf},a1);
  s1:=simplify(sl[2]);
```

$$eq := \left(a1\, r - \frac{5}{8}\, a1^5\right) \cos(x) - \frac{1}{16}\, a1^5 \cos(5\, x) - \frac{5}{16}\, a1^5 \cos(3\, x)$$

$$cf := -\frac{1}{8}\, a1\, (5\, a1^4 - 8\, r)$$

$$sl := \{a1 = 0\}, \left\{a1 = \frac{1}{5}\sqrt{10}\sqrt{\sqrt{10}\sqrt{r}}\right\}, \left\{a1 = -\frac{1}{5}\sqrt{10}\sqrt{\sqrt{10}\sqrt{r}}\right\}, \left\{a1 = \frac{1}{5}\, \mathrm{I}\sqrt{10}\sqrt{\sqrt{10}\sqrt{r}}\right\}, \left\{a1 = -\frac{1}{5}\, \mathrm{I}\sqrt{10}\sqrt{\sqrt{10}\sqrt{r}}\right\}$$

$$s1 := \left\{a1 = \frac{1}{5}\, 10^{3/4}\, r^{1/4}\right\}$$

```
> eq:=collect(combine(subs(s1,N(a1*cos(x)+a3*cos(3*x)))),{cos(x),
  cos(3*x)}):
  cf:=simplify(coeff(eq,cos(3*x),1));
  sl:=solve({-64*a3-1/10*10^(3/4)*r^(5/4)},a3);
  s1:=s1 union sl;
```

$$cf := -\frac{3}{2}\,10^{1/4}\,r^{3/4}\,a3^2 - 64\,a3 - \frac{1}{10}\,10^{3/4}\,r^{5/4} - \frac{5}{8}\,a3^5 - 2\,r\,a3 - \frac{3}{2}\sqrt{10}\,\sqrt{r}\,a3^3$$

$$sl := \left\{a3 = -\frac{1}{640}\,10^{3/4}\,r^{5/4}\right\}$$

$$s1 := \left\{a1 = \frac{1}{5}\,10^{3/4}\,r^{1/4},\ a3 = -\frac{1}{640}\,10^{3/4}\,r^{5/4}\right\}$$

```
> eq:=collect(combine(subs(s1,N(a1*cos(x)+a3*cos(3*x)+a5*cos(5*x)))
  ),{cos(x),cos(3*x),cos(5*x)}):
  cf:=simplify(coeff(eq,cos(5*x),1));
  sl:=solve({-1/50*10^(3/4)*r^(5/4)-576*a5},a5);
```

$$cf := \frac{9}{256}\,10^{1/4}\,r^{7/4}\,a5^2 - \frac{3}{2}\sqrt{10}\,\sqrt{r}\,a5^3 - \frac{9}{32768}\,10^{1/4}\,r^{11/4}\,a5^2 - \frac{5}{8}\,a5^5 - \frac{1}{671088640}\,10^{3/4}\,r^{21/4} + \frac{1}{1048576}\,r^4\,a5 + \frac{1}{320}\,10^{3/4}\,r^{9/4} + \frac{3}{10485760}\,10^{3/4}\,r^{17/4} + \frac{1}{32}\,r^2\,a5 - \frac{3}{81920}\,10^{3/4}\,r^{13/4} - \frac{3}{32768}\sqrt{10}\,r^{5/2}\,a5^3 - \frac{3}{268435456}\,r^5\,a5 - 576\,a5 - 2\,r\,a5 - \frac{1}{50}\,10^{3/4}\,r^{5/4} - \frac{3}{4096}\,r^3\,a5$$

$$sl := \left\{a5 = -\frac{1}{28800}\,10^{3/4}\,r^{5/4}\right\}$$

> **# In a nicer form:** $\mathbf{a1 = \left(\frac{8}{5}\right)^{\frac{1}{4}} r^{\frac{1}{4}} = 1.125\, r^{\frac{1}{4}},\ a3 = -\frac{10^{-\frac{1}{4}}}{64}\, r^{\frac{5}{4}} = -0.00878\, r^{\frac{5}{4}},\ a3 = \frac{10^{-\frac{1}{4}}}{2880}\, r^{\frac{5}{4}} = -0.000195\, r^{\frac{5}{4}}}$

(c)

Square state:

```
> f(x,y):=cos(x)+cos(y);
  eq:=collect(combine(N(a1*f(x,y))),{cos(x),cos(y)}):
  cf:=factor(coeff(eq,cos(x),1));
  #cf:=factor(coeff(eq,cos(y),1));
```

```
sl:=solve({cf},a1);
s2:=sl[2]; evalf(s2);
```

$$f(x, y) := \cos(x) + \cos(y)$$

$$cf := -\frac{1}{4}\, a1\, (25\, a1^4 - 4\, r)$$

$$sl := \{a1 = 0\}, \left\{a1 = \frac{1}{5}\sqrt{10}\, r^{1/4}\right\}, \left\{a1 = -\frac{1}{5}\sqrt{10}\, r^{1/4}\right\}, \left\{a1 = \frac{1}{5}\, \mathrm{I}\sqrt{10}\, r^{1/4}\right\}, \left\{a1 = -\frac{1}{5}\, \mathrm{I}\sqrt{10}\, r^{1/4}\right\}$$

$$s2 := \left\{a1 = \frac{1}{5}\sqrt{10}\, r^{1/4}\right\}$$

$$\{a1 = 0.6324555320\, r^{1/4}\}$$

> # **In a nicer form: a1** $= \left(\frac{2}{5}\right)^{\frac{1}{2}} r^{\frac{1}{4}}$

(d)

Hexagonal state:

```
> f(x,y):=cos(x)+cos(-1/2*x+sqrt(3)/2*y)+cos(-1/2*x-sqrt(3)/2*y);
eq:=collect(combine(N(a1*f(x,y))),{cos(x),cos(-1/2*x+sqrt(3)/2*
y),cos(-1/2*x-sqrt(3)/2*y)}):
cf:=factor(coeff(eq,cos(x),1));
#cf:=factor(coeff(eq,cos(-1/2*x+sqrt(3)/2*y),1));
#cf:=factor(coeff(eq,cos(-1/2*x-sqrt(3)/2*y),1));
sl:=solve({cf},a1);
s3:=sl[2];
simplify(s3,radi);
```

$$f(x, y) := \cos(x) + \cos\left(\frac{1}{2}\, x - \frac{1}{2}\sqrt{3}\, y\right) + \cos\left(\frac{1}{2}\, x + \frac{1}{2}\sqrt{3}\, y\right)$$

$$cf := -\frac{1}{4}\, a1\, (85\, a1^4 - 4\, r)$$

$$sl := \{a1 = 0\}, \left\{a1 = \frac{1}{85}\sqrt{170}\sqrt{\sqrt{85}\sqrt{r}}\right\}, \left\{a1 = -\frac{1}{85}\sqrt{170}\sqrt{\sqrt{85}\sqrt{r}}\right\}, \left\{a1 = \frac{1}{85}\, \mathrm{I}\sqrt{170}\sqrt{\sqrt{85}\sqrt{r}}\right\}, \left\{a1 = -\frac{1}{85}\, \mathrm{I}\sqrt{170}\sqrt{\sqrt{85}\sqrt{r}}\right\}$$

$$s3 := \left\{a1 = \frac{1}{85}\sqrt{170}\sqrt{\sqrt{85}\sqrt{r}}\right\}$$

$$\left\{a1 = \frac{1}{85}\sqrt{170}\sqrt{\sqrt{85}\sqrt{r}}\right\}$$

```
> # In a nicer form: a1=(4/85)^(1/4) r^(1/4)
```

$$a1=\left(\frac{4}{85}\right)^{\frac{1}{4}} r^{\frac{1}{4}}$$

(e)

Cross-roll instability of stripes:

```
> eq:=subs(s1,collect(combine(coeff(N(a1*cos(x)+epsilon*(alpha*cos
  (x)+beta*cos(Q*sin(phi)*x+Q*cos(phi)*y))),epsilon,1),trig),{cos
  (x),cos(Q*sin(phi)*x+Q*cos(phi)*y)}));
```

$$eq := -4\cos(x)\,\alpha\, r+\left(-Q^4\beta+2\,Q^2\beta-2\,\beta\, r-\beta\right)\cos(Q\sin(\phi)\, x+Q\cos(\phi)\, y) -\frac{1}{2}\, r\beta\cos(Q\sin(\phi)\, x+Q\cos(\phi)\, y-4\, x)-\frac{1}{2}\, r\beta\cos(Q\sin(\phi)\, x +Q\cos(\phi)\, y+4\, x)-2\, r\beta\cos(Q\sin(\phi)\, x+Q\cos(\phi)\, y-2\, x) -2\, r\beta\cos(Q\sin(\phi)\, x+Q\cos(\phi)\, y+2\, x)-\frac{5}{2}\, r\alpha\cos(3\, x)-\frac{1}{2}\, r\alpha\cos(5\, x)$$

```
> A11:=factor(coeff(coeff(eq,cos(x),1),alpha,1)):
  A12:=factor(coeff(coeff(eq,cos(x),1),beta,1)):
  A21:=factor(coeff(coeff(eq,cos(Q*sin(phi)*x+Q*cos(phi)*y),1),
  alpha,1)):
  A22:=factor(coeff(coeff(eq,cos(Q*sin(phi)*x+Q*cos(phi)*y),1),
  beta,1)):
  A:=matrix(2,2,[A11,A12,A21,A22]);
  e1:=eigenvectors(A);
```

$$A := \begin{bmatrix} -4\, r & 0 \\ 0 & -Q^4+2\,Q^2-2\, r-1 \end{bmatrix}$$

$$e1 := \left[-Q^4+2\,Q^2-2\, r-1,\ 1,\ \left\{\left[\begin{matrix} 0 & 1 \end{matrix}\right]\right\}\right],\ \left[-4\, r,\ 1,\ \left\{\left[\begin{matrix} 1 & 0 \end{matrix}\right]\right\}\right]$$

```
> # The eigenvalues are -4 r and -2 r-(1-Q^2)^2 <-2 r
```

$$-4\,\mathrm{r} \text{ and } -2\,\mathrm{r}-\left(1-\mathrm{Q}^2\right)^2 < -2\,\mathrm{r}$$

Both eigenvalues are strictly negative (for any Q and ϕ), so stripes are cross-roll stable.

(f)

Linear stability of squares towards stripes:

```
> eq:=subs(s2,collect(combine(coeff(N(a1*(cos(x)+cos(y))+epsilon*
  (alpha*cos(x)+beta*cos(y))),epsilon,1),trig),{cos(x),cos(y)})):
> A11:=factor(coeff(coeff(eq,cos(x),1),alpha,1)):
  A12:=factor(coeff(coeff(eq,cos(x),1),beta,1)):
  A21:=factor(coeff(coeff(eq,cos(y),1),alpha,1)):
  A22:=factor(coeff(coeff(eq,cos(y),1),beta,1)):
  A:=matrix(2,2,[A11,A12,A21,A22]);
  e1:=eigenvectors(A);
```

$$A := \begin{bmatrix} -\frac{8}{5}r & -\frac{12}{5}r \\ -\frac{12}{5}r & -\frac{8}{5}r \end{bmatrix}$$

$$e1 := \left[-4r, 1, \{[1 \;\; 1]\}\right], \left[\frac{4}{5}r, 1, \{[-1 \;\; 1]\}\right]$$

> # The eigenvalues are -4 r and $\frac{4}{5}$ r

The second eigenvalue is positive, so squares unstable towards stripes.

(g)

Linear stability of hexagons towards stripes:

```
> eq:=collect(subs({cos(1/2*x-sqrt(3)/2*y)=cos(-1/2*x+sqrt(3)/2*y),
  cos(1/2*x+sqrt(3)/2*y)=cos(-1/2*x-sqrt(3)/2*y)},subs(s3,combine
  (coeff(N(a1*(cos(x)+cos(-1/2*x+sqrt(3)/2*y)+cos(-1/2*x-sqrt(3)/2*
  y))+epsilon*(alpha*cos(x)+beta*cos(-1/2*x+sqrt(3)/2*y)+gamma*cos
  (-1/2*x-sqrt(3)/2*y))),epsilon,1),trig))),{cos(x),cos(-1/2*x+sqrt
  (3)/2*y),cos(-1/2*x-sqrt(3)/2*y)}):
> A11:=factor(coeff(coeff(eq,cos(x),1),alpha,1)):
  A12:=factor(coeff(coeff(eq,cos(x),1),beta,1)):
  A13:=factor(coeff(coeff(eq,cos(x),1),gamma,1)):
  A21:=factor(coeff(coeff(eq,cos(-1/2*x+sqrt(3)/2*y),1),alpha,1)):
  A22:=factor(coeff(coeff(eq,cos(-1/2*x+sqrt(3)/2*y),1),beta,1)):
  A23:=factor(coeff(coeff(eq,cos(-1/2*x+sqrt(3)/2*y),1),gamma,1)):
  A31:=factor(coeff(coeff(eq,cos(-1/2*x-sqrt(3)/2*y),1),alpha,1)):
  A32:=factor(coeff(coeff(eq,cos(-1/2*x-sqrt(3)/2*y),1),beta,1)):
  A33:=factor(coeff(coeff(eq,cos(-1/2*x-sqrt(3)/2*y),1),gamma,1)):
  A:=matrix(3,3,[A11,A12,A13,A21,A22,A23,A31,A32,A33]);
  e1:=eigenvectors(A);
```

$$A := \begin{bmatrix} -\frac{14}{17}r & -\frac{27}{17}r & -\frac{27}{17}r \\ -\frac{27}{17}r & -\frac{14}{17}r & -\frac{27}{17}r \\ -\frac{27}{17}r & -\frac{27}{17}r & -\frac{14}{17}r \end{bmatrix}$$

$$e1 := \left[\frac{13}{17}r, 2, \{[-1 \;\; 0 \;\; 1], [-1 \;\; 1 \;\; 0]\}\right], \left[-4r, 1, \{[1 \;\; 1 \;\; 1]\}\right]$$

> # The eigenvalues are -4 r and $\frac{13}{17}$ r

Again, the second eigenvalue is positive, so hexagons are unstable towards stripes.

(h)

At onset the stripes will be selected as the only stable state.

(i)

Stripe solution with q different from unity:

```
> eq:=collect(combine(N(a1*cos(q*x))),{cos(x)});
  cf:=factor(coeff(eq,cos(q*x),1));
  sl:=solve(subs(a1^4=s,cf),s):
  s1:={a1=(sl)^(1/4)};
  r0:=solve(subs(s1,a1),r);
```

$$eq := a1\cos(q\,x)\,r - a1\cos(q\,x) + 2\,a1\cos(q\,x)\,q^2 - a1\cos(q\,x)\,q^4 - \frac{1}{16}\,a1^5\cos(5\,q\,x) - \frac{5}{16}\,a1^5\cos(3\,q\,x) - \frac{5}{8}\,a1^5\cos(q\,x)$$

$$cf := -\frac{1}{8}\,a1\,(5\,a1^4 + 8\,q^4 - 16\,q^2 - 8\,r + 8)$$

$$s1 := \left\{a1 = \left(-\frac{8}{5}\,q^4 + \frac{16}{5}\,q^2 + \frac{8}{5}\,r - \frac{8}{5}\right)^{1/4}\right\}$$

Warning, solve may be ignoring assumptions on the input variables.

$$r0 := q^4 - 2\,q^2 + 1$$

> # **In a nicer form: and** $\mathbf{a1} = \left(\frac{8}{5}\cdot\left(r - (1-q^2)^2\right)\right)^{\frac{1}{4}}$.

(j)

Cross-roll instability of stripes:

```
> eq:=subs(s1,collect(combine(coeff(N(a1*cos(q*x)+epsilon*(alpha*
  cos(q*x)+beta*cos(Q*sin(phi)*x+Q*cos(phi)*y))),epsilon,1),trig),
  {cos(q*x),cos(Q*sin(phi)*x+Q*cos(phi)*y)})):
> A11:=factor(coeff(coeff(eq,cos(q*x),1),alpha,1)):
  A12:=factor(coeff(coeff(eq,cos(q*x),1),beta,1)):
  A21:=factor(coeff(coeff(eq,cos(Q*sin(phi)*x+Q*cos(phi)*y),1),
  alpha,1)):
  A22:=factor(coeff(coeff(eq,cos(Q*sin(phi)*x+Q*cos(phi)*y),1),
  beta,1)):
  A:=matrix(2,2,[A11,A12,A21,A22]);
  e1:=eigenvectors(A);
```

$$A := \begin{bmatrix} 4\,q^4 - 8\,q^2 - 4\,r + 4 & 0 \\ 0 & -Q^4 + 3\,q^4 + 2\,Q^2 - 6\,q^2 - 2\,r + 2 \end{bmatrix}$$

$$e1 := \left[-Q^4 + 3\,q^4 + 2\,Q^2 - 6\,q^2 - 2\,r + 2,\ 1,\ \{[\,0\ \ 1\,]\}\right],\ \left[4\,q^4 - 8\,q^2 - 4\,r + 4,\ 1,\ \{[\,1\ \ 0\,]\}\right]$$

> # **The eigenvalues are** $\mathbf{e1} = -4\,(\mathbf{r} - (1-\mathbf{q}^2)^2) < 0$ **and**

$$e2 := \text{expand}\left(-2\,r + 3\left(1-q^2\right)^2 - \left(1-Q^2\right)^2\right);$$

$$e2 := -Q^4 + 3\,q^4 + 2\,Q^2 - 6\,q^2 - 2\,r + 2 \quad \mathbf{(1)}$$

The first eigenvalue is non-positive where the stripe solution exists. The second eigenvalue achieves a (positive) maximum at Q=1, which gives the neutral stability curve given by

```
> r1:=simplify(solve(subs(Q=1,e2),r));
```

Warning, solve may be ignoring assumptions on the input variables.

$$r1 := \frac{3}{2}\,q^4 - 3\,q^2 + \frac{3}{2}$$

In particular, near q=1 we get:

```
> r1a:=convert(series(r1,q=1,3),polynom);
```

$$r1a := 6\,(q-1)^2$$

(k)

Stability with respect to zigzag instability:

```
> eq:=subs(s1,collect(expand(combine(coeff(N(a1*cos(q*x)+epsilon*
  (alpha*cos(q*x)*exp(I*Q*y)+beta*sin(q*x)*exp(I*Q*y))),epsilon,1),
  trig)),{cos(q*x),sin(q*y)}));
```

$$eq := -5\,e^{IQy}\left(-\frac{8}{5}\,q^4 + \frac{16}{5}\,q^2 + \frac{8}{5}\,r - \frac{8}{5}\right)\alpha\cos(q\,x)^5 - 5\,e^{IQy}\left(-\frac{8}{5}\,q^4 + \frac{16}{5}\,q^2 + \frac{8}{5}\,r - \frac{8}{5}\right)\beta\sin(q\,x)\cos(q\,x)^4 + \left(-e^{IQy}\,Q^4\,\alpha - 2\,e^{IQy}\,Q^2\,\alpha\,q^2 - e^{IQy}\,\alpha\,q^4 + 2\,e^{IQy}\,Q^2\,\alpha + 2\,e^{IQy}\,\alpha\,q^2 + e^{IQy}\,\alpha\,r - e^{IQy}\,\alpha\right)\cos(q\,x) - \beta\sin(q\,x)\,Q^4\,e^{IQy} - 2\,\beta\sin(q\,x)\,q^2\,Q^2\,e^{IQy} - \beta\sin(q\,x)\,q^4\,e^{IQy} + 2\,\beta\sin(q\,x)\,Q^2\,e^{IQy} + 2\,\beta\sin(q\,x)\,q^2\,e^{IQy} + e^{IQy}\sin(q\,x)\,\beta\,r - \beta\sin(q\,x)\,e^{IQy}$$

```
> eq1:=eval(int(eq*cos(q*x),x=0..2*Pi/q))*q/Pi/exp(I*Q*y):
  eq2:=eval(int(eq*sin(q*x),x=0..2*Pi/q))*q/Pi/exp(I*Q*y):
  A11:=factor(coeff(eq1,alpha,1)):
  A12:=factor(coeff(eq1,beta,1)):
  A21:=factor(coeff(eq2,alpha,1)):
  A22:=factor(coeff(eq2,beta,1)):
  A:=matrix(2,2,[A11,A12,A21,A22]);
  e1:=eigenvectors(A);
```

$$A := \begin{bmatrix} -Q^4 - 2\,Q^2\,q^2 + 4\,q^4 + 2\,Q^2 - 8\,q^2 - 4\,r + 4 & 0 \\ 0 & -Q^2\left(Q^2 + 2\,q^2 - 2\right) \end{bmatrix}$$

$$e1 := \left[-Q^4 - 2\,Q^2\,q^2 + 2\,Q^2,\ 1,\ \left\{\begin{bmatrix} 0 & 1 \end{bmatrix}\right\}\right],\ \left[-Q^4 - 2\,Q^2\,q^2 + 4\,q^4 + 2\,Q^2 - 8\,q^2 - 4\,r + 4,\ 1,\ \left\{\begin{bmatrix} 1 & 0 \end{bmatrix}\right\}\right]$$

> # The eigenvalues are: $e1 = -Q^4 + 2(1 - q^2)Q^2$ and $e2 = -4\cdot(r - (1-q^2)^2) - Q^4 + 2(1 - q^2)Q^2 < e1$

The maximum of both expressions over all q and Q is positive for q<1 and non-positive otherwise, so the stripes are zigzag-stable for q>1 and unstable for q<1; so the stability boundary is q=1.

(l)

Same for the Eckhaus stability:

```
> eq:=subs(s1,collect((combine(coeff(N(a1*cos(q*x)+epsilon*(alpha*
  cos(q*x)*exp(I*Q*x)+beta*sin(q*x)*exp(I*Q*x))),epsilon,1),trig)),
  {cos(q*x),sin(q*y)}));
```

$$eq := \Big(2\,\alpha\,Q^2\,e^{IQx} - \alpha\,Q^4\,e^{IQx} + 4\,I\,\beta\,q^3\,Q\,e^{IQx} + 2\,\alpha\,q^2\,e^{IQx} - \alpha\,q^4\,e^{IQx} - 4\,I\,\beta\,q\,Q\,e^{IQx} + e^{IQx}\,\alpha\,r + 4\,I\,\beta\,q\,Q^3\,e^{IQx} - 6\,\alpha\,q^2\,Q^2\,e^{IQx} - \frac{25}{8}\,e^{IQx}\Big(-\frac{8}{5}\,q^4 + \frac{16}{5}\,q^2 + \frac{8}{5}\,r - \frac{8}{5}\Big)\alpha - \alpha\,e^{IQx}\Big)\cos(q\,x) - \beta\sin(q\,x)\,e^{IQx} + e^{IQx}\sin(q\,x)\,\beta\,r - 4\,I\,\alpha\sin(q\,x)\,q^3\,Q\,e^{IQx} - 6\,\beta\sin(q\,x)\,q^2\,Q^2\,e^{IQx} + 2\,\beta\sin(q\,x)\,q^2\,e^{IQx} + 2\,\beta\sin(q\,x)\,Q^2\,e^{IQx} - \frac{5}{16}\,e^{IQx}\Big(-\frac{8}{5}\,q^4 + \frac{16}{5}\,q^2 + \frac{8}{5}\,r - \frac{8}{5}\Big)\alpha\cos(5\,q\,x) - \beta\sin(q\,x)\,q^4\,e^{IQx} - \beta\sin(q\,x)\,Q^4\,e^{IQx} - 4\,I\,\alpha\sin(q\,x)\,q\,Q^3\,e^{IQx} - \frac{15}{16}\,e^{IQx}\Big(-\frac{8}{5}\,q^4 + \frac{16}{5}\,q^2 + \frac{8}{5}\,r - \frac{8}{5}\Big)\beta\sin(3\,q\,x) + 4\,I\,\alpha\sin(q\,x)\,q\,Q\,e^{IQx} - \frac{25}{16}\,e^{IQx}\Big(-\frac{8}{5}\,q^4 + \frac{16}{5}\,q^2 + \frac{8}{5}\,r - \frac{8}{5}\Big)\alpha\cos(3\,q\,x) - \frac{5}{16}\,e^{IQx}\Big(-\frac{8}{5}\,q^4 + \frac{16}{5}\,q^2 + \frac{8}{5}\,r - \frac{8}{5}\Big)\beta\sin(5\,q\,x) - \frac{5}{8}\,e^{IQx}\sin(q\,x)\Big(-\frac{8}{5}\,q^4 + \frac{16}{5}\,q^2 + \frac{8}{5}\,r - \frac{8}{5}\Big)\beta$$

```
> A11:=factor(coeff(coeff(eq,cos(q*x),1),alpha,1))/exp(I*Q*x);
  A12:=factor(coeff(coeff(eq,cos(q*x),1),beta,1))/exp(I*Q*x);
  A21:=factor(coeff(coeff(eq,sin(q*x),1),alpha,1))/exp(I*Q*x);
  A22:=factor(coeff(coeff(eq,sin(q*x),1),beta,1))/exp(I*Q*x);
  A:=matrix(2,2,[A11,A12,A21,A22]);
  d1:=factor(det(A));
```

$$A11 := -Q^4 - 6\,Q^2\,q^2 + 4\,q^4 + 2\,Q^2 - 8\,q^2 - 4\,r + 4$$

$$A12 := 4\,I\,q\,Q\,(Q^2 + q^2 - 1)$$

$$A21 := -4\,I\,q\,Q\,(Q^2 + q^2 - 1)$$

$$A22 := -Q^2\,(Q^2 + 6\,q^2 - 2)$$

$$A := \begin{bmatrix} -Q^4 - 6\,Q^2\,q^2 + 4\,q^4 + 2\,Q^2 - 8\,q^2 - 4\,r + 4 & 4\,\mathrm{I}\,q\,Q\,(Q^2 + q^2 - 1) \\ -4\,\mathrm{I}\,q\,Q\,(Q^2 + q^2 - 1) & -Q^2\,(Q^2 + 6\,q^2 - 2) \end{bmatrix}$$

$$d1 := Q^2\,(Q^6 - 4\,Q^4\,q^2 - 40\,q^6 - 4\,Q^4 + 16\,Q^2\,q^2 + 88\,q^4 + 4\,Q^2\,r + 24\,q^2\,r - 56\,q^2 - 8\,r + 8)$$

The neutral stability curve can be computed by setting Q=0 (this is a long-wavelength instability) and setting the determinant to zero:

```
> r2:=solve(subs(Q=0,factor(d1/Q^2)),r);
```

Warning, solve may be ignoring assumptions on the input variables.

$$r2 := \frac{5\,q^6 - 11\,q^4 + 7\,q^2 - 1}{3\,q^2 - 1}$$

Near q=1 we get:

```
> r2a:=subs(k=q-1,convert(series(subs(q=1+k,r2),k,3),polynom));
```

$$r2a := 8\,(q-1)^2$$

so the Eckhaus instability precedes the cross-roll. The stability baloon is therefore limited by q=1 on the left and r=r2(q) on the right:

```
> with(plots):
  pl:=plot([r0,r1,r2,Heaviside(q-1)],q=0.9..1.2,color=["red",
  "green","blue","black"],legend=["Neutral stability","Cross Roll",
  "Eckhaus","Zigzag"]): display(pl,view=[0.9..1.2,0..0.3]);
```



>

Problem 2

```
> assume(g>0):
  L:=u->diff(u,x,x)+diff(u,y,y);
  N:=u->(r-1)*u-2*L(u)-L(L(u))-g*u^2-u^3;
```

$$L := u \rightarrow \frac{\partial^2}{\partial x^2} u + \frac{\partial^2}{\partial y^2} u$$

$$N := u \rightarrow (r-1)\, u - 2\, L(u) - L(L(u)) - g\, u^2 - u^3$$

(a)

We have computed earlier (assignment 7) that the stripes have an amplitude

```
> s1:={a1=2/3*(3*r)^(1/2)};
```

$$s1 := \left\{ a1 = \frac{2}{3} \sqrt{3} \sqrt{r} \right\}$$

(b)

Square state (we have already found that the 0th and 2nd harmonic contributions are of higher order r compared to the 1st harmonic which scales as r^(1/2), so ignore them):

```
> f(x,y):=cos(x)+cos(y);
  eq:=collect(combine(N(a1*f(x,y))),{cos(x),cos(y)});
  cf1:=factor(coeff(eq,cos(x),1));
  sl:=solve({cf1},{a1});
  s2:=sl[2];
```

$$f(x, y) := \cos(x) + \cos(y)$$

$$eq := \left(-\frac{9}{4}\, a1^3 + a1\, r\right)\cos(x) + \left(-\frac{9}{4}\, a1^3 + a1\, r\right)\cos(y) - \frac{1}{4}\, a1^3 \cos(3\,x) - \frac{1}{2}\, g\, a1^2 \cos(2\,y) - g\, a1^2 \cos(x-y) - g\, a1^2 \cos(x+y) - \frac{1}{2}\, g\, a1^2 \cos(2\,x) - \frac{3}{4}\, a1^3 \cos(-y+2\,x) - g\, a1^2 - \frac{3}{4}\, a1^3 \cos(x-2\,y) - \frac{1}{4}\, a1^3 \cos(3\,y) - \frac{3}{4}\, a1^3 \cos(y+2\,x) - \frac{3}{4}\, a1^3 \cos(x+2\,y)$$

$$cf1 := -\frac{1}{4}\, a1\,(9\, a1^2 - 4\, r)$$

$$sl := \{a1 = 0\},\ \left\{a1 = \frac{2}{3}\sqrt{r}\right\},\ \left\{a1 = -\frac{2}{3}\sqrt{r}\right\}$$

$$s2 := \left\{a1 = \frac{2}{3}\sqrt{r}\right\}$$

(c)

Hexagonal state (again ignore the constant term as higher order correction):

```
> f(x,y):=cos(x)+cos(-1/2*x+sqrt(3)/2*y)+cos(-1/2*x-sqrt(3)/2*y);
  eq:=collect(combine(N(a1*f(x,y))),{cos(x),cos(-1/2*x+sqrt(3)/2*
  y),cos(-1/2*x-sqrt(3)/2*y)}):
  cf1:=factor(coeff(eq,cos(x),1));
  sl:=solve({cf1},a1);
  s3p:=sl[2];
  a1=series(subs(s3p,a1),r,2);
  s3m:=sl[3];
  a1=series(subs(s3m,a1),r,2);
```

$$f(x, y) := \cos(x) + \cos\left(\frac{1}{2}\, x - \frac{1}{2}\sqrt{3}\, y\right) + \cos\left(\frac{1}{2}\, x + \frac{1}{2}\sqrt{3}\, y\right)$$

$$cf1 := -\frac{1}{4}\, a1\,(15\, a1^2 + 4\, a1\, g - 4\, r)$$

$$sl := \{a1 = 0\},\ \left\{a1 = -\frac{2}{15}\, g + \frac{2}{15}\sqrt{g^2 + 15\, r}\right\},\ \left\{a1 = -\frac{2}{15}\, g - \frac{2}{15}\sqrt{g^2 + 15\, r}\right\}$$

$$s3p := \left\{a1 = -\frac{2}{15}\,g + \frac{2}{15}\sqrt{g^2 + 15\,r}\right\}$$

$$a1 = \frac{1}{g}\,r + \mathrm{O}(r^2)$$

$$s3m := \left\{a1 = -\frac{2}{15}\,g - \frac{2}{15}\sqrt{g^2 + 15\,r}\right\}$$

$$a1 = -\frac{4}{15}\,g - \frac{1}{g}\,r + \mathrm{O}(r^2)$$

```
> g0:=0.1;
  plot(subs(g=g0,[abs(subs(s3p,a1)),abs(subs(s3m,a1)),subs(s1,a1),
  subs(s2,a1)]),r=-g0^2/15..2*g0^2/15,color=["red","green","blue",
  "black"],legend=["- hexagons","+ hexagons","stripes","squares"]);
```

$$g0 := 0.1$$



The amplitude of the amplitude of the "-" hexagons (s3p) grows linearly with distance to the bifurcation, indicating a transcritical bifurcation. The amplitude of the "+" hexagons is O(1) at r=0, it is created at r=-

g^2/15<0 via a saddle-node bifurcation (sub-critical bifurcation). The stripes and squares are created via a supercritical pitchfork bifurcation at r=0.

(d)

Cross-roll instability of stripes:

```
> eq:=subs(s1,collect(combine(coeff(N(a1*cos(x)+epsilon*(alpha*cos
  (x)+beta*cos(Q*sin(phi)*x+Q*cos(phi)*y))),epsilon,1),trig),{cos
  (x),cos(Q*sin(phi)*x+Q*cos(phi)*y)}));
```

$$eq := -2\cos(x)\,\alpha\, r + \left(-Q^4\beta + 2\,Q^2\beta - \beta\, r - \beta\right)\cos(Q\sin(\phi)\,x + Q\cos(\phi)\,y) - \frac{2}{3}\sqrt{3}\sqrt{r}\,\beta\, g\cos(-x + Q\sin(\phi)\,x + Q\cos(\phi)\,y) - \frac{2}{3}\sqrt{3}\sqrt{r}\,\beta\, g\cos(x + Q\sin(\phi)\,x + Q\cos(\phi)\,y) - \frac{2}{3}\sqrt{3}\sqrt{r}\,\alpha\, g - \frac{2}{3}\sqrt{3}\sqrt{r}\,\alpha\, g\cos(2\,x) - r\alpha\cos(3\,x) - r\beta\cos(Q\sin(\phi)\,x + Q\cos(\phi)\,y - 2\,x) - r\beta\cos(Q\sin(\phi)\,x + Q\cos(\phi)\,y + 2\,x)$$

```
> A11:=factor(coeff(coeff(eq,cos(x),1),alpha,1)):
  A12:=factor(coeff(coeff(eq,cos(x),1),beta,1)):
  A21:=factor(coeff(coeff(eq,cos(Q*sin(phi)*x+Q*cos(phi)*y),1),
  alpha,1)):
  A22:=factor(coeff(coeff(eq,cos(Q*sin(phi)*x+Q*cos(phi)*y),1),
  beta,1)):
  A:=matrix(2,2,[A11,A12,A21,A22]);
  e1:=eigenvectors(A);
```

$$A := \begin{bmatrix} -2\,r & 0 \\ 0 & -Q^4 + 2\,Q^2 - r - 1 \end{bmatrix}$$

$$e1 := \left[-Q^4 + 2\,Q^2 - r - 1,\ 1,\ \left\{\begin{bmatrix} 0 & 1 \end{bmatrix}\right\}\right],\ \left[-2\,r,\ 1,\ \left\{\begin{bmatrix} 1 & 0 \end{bmatrix}\right\}\right]$$

Both eigenvalues are strictly negative (for any Q and phi), so stripes are cross-roll stable.

(e)

Linear stability of squares towards stripes:

```
> eq:=subs(s2,collect(combine(coeff(N(a1*(cos(x)+cos(y))+epsilon*
  (alpha*cos(x)+beta*cos(y))),epsilon,1),trig),{cos(x),cos(y)}));
```

$$eq := \left(-\frac{4}{3}\beta\, r - \frac{2}{3}\alpha\, r\right)\cos(x) + \left(-\frac{2}{3}\beta\, r - \frac{4}{3}\alpha\, r\right)\cos(y) - \frac{2}{3}\sqrt{r}\,\beta\, g - \frac{2}{3}\sqrt{r}\,\alpha\, g - \frac{2}{3}\sqrt{r}\,\alpha\, g\cos(2\,x) - \frac{1}{3}\, r\beta\cos(-y + 2\,x) - \frac{1}{3}\, r\beta\cos(y + 2\,x) - \frac{1}{3}\, r\beta\cos(3\,y) - \frac{2}{3}\, r\alpha\cos(y + 2\,x) - \frac{1}{3}\, r\alpha\cos(x - 2\,y) - \frac{1}{3}\, r\alpha\cos(x$$

$$+2y) - \frac{2}{3}\sqrt{r}\,\beta\,g\cos(x+y) - \frac{2}{3}\,r\beta\cos(x-2y) - \frac{2}{3}\,r\beta\cos(x+2y)$$

$$-\frac{2}{3}\sqrt{r}\,\alpha\,g\cos(x+y) - \frac{1}{3}\,r\alpha\cos(3x) - \frac{2}{3}\sqrt{r}\,\alpha\,g\cos(x-y) - \frac{2}{3}\,r\alpha\cos($$

$$-y+2x) - \frac{2}{3}\sqrt{r}\,\beta\,g\cos(2y) - \frac{2}{3}\sqrt{r}\,\beta\,g\cos(x-y)$$

```
> A11:=factor(coeff(coeff(eq,cos(x),1),alpha,1)):
  A12:=factor(coeff(coeff(eq,cos(x),1),beta,1)):
  A21:=factor(coeff(coeff(eq,cos(y),1),alpha,1)):
  A22:=factor(coeff(coeff(eq,cos(y),1),beta,1)):
  A:=matrix(2,2,[A11,A12,A21,A22]);
  e1:=eigenvectors(A);
```

$$A := \begin{bmatrix} -\frac{2}{3}\,r & -\frac{4}{3}\,r \\ -\frac{4}{3}\,r & -\frac{2}{3}\,r \end{bmatrix}$$

$$e1 := \left[\frac{2}{3}\,r,\,1,\,\left\{\left[\begin{array}{cc} -1 & 1 \end{array}\right]\right\}\right],\,\left[-2\,r,\,1,\,\left\{\left[\begin{array}{cc} 1 & 1 \end{array}\right]\right\}\right]$$

One eigenvalue is positive, so squares unstable towards stripes.

(f)

Linear stability of hexagons towards stripes:

```
> eq:=collect(subs({cos(1/2*x-sqrt(3)/2*y)=cos(-1/2*x+sqrt(3)/2*y),
  cos(1/2*x+sqrt(3)/2*y)=cos(-1/2*x-sqrt(3)/2*y)},subs(s3p,combine
  (coeff(N(a1*(cos(x)+cos(-1/2*x+sqrt(3)/2*y)+cos(-1/2*x-sqrt(3)/2*
  y))+epsilon*(alpha*cos(x)+beta*cos(-1/2*x+sqrt(3)/2*y)+gamma*cos
  (-1/2*x-sqrt(3)/2*y))),epsilon,1),trig))),{cos(x),cos(-1/2*x+sqrt
  (3)/2*y),cos(-1/2*x-sqrt(3)/2*y)}):
```

```
> A11:=factor(coeff(coeff(eq,cos(x),1),alpha,1)):
  A12:=factor(coeff(coeff(eq,cos(x),1),beta,1)):
  A13:=factor(coeff(coeff(eq,cos(x),1),gamma,1)):
  A21:=factor(coeff(coeff(eq,cos(-1/2*x+sqrt(3)/2*y),1),alpha,1)):
  A22:=factor(coeff(coeff(eq,cos(-1/2*x+sqrt(3)/2*y),1),beta,1)):
  A23:=factor(coeff(coeff(eq,cos(-1/2*x+sqrt(3)/2*y),1),gamma,1)):
  A31:=factor(coeff(coeff(eq,cos(-1/2*x-sqrt(3)/2*y),1),alpha,1)):
  A32:=factor(coeff(coeff(eq,cos(-1/2*x-sqrt(3)/2*y),1),beta,1)):
  A33:=factor(coeff(coeff(eq,cos(-1/2*x-sqrt(3)/2*y),1),gamma,1)):
  A:=matrix(3,3,[A11,A12,A13,A21,A22,A23,A31,A32,A33]);
  e1:=eigenvectors(A);
```

$$A := \left[\left[-\frac{2}{5}\,r - \frac{14}{75}\,g^2 + \frac{14}{75}\,g\sqrt{g^2+15\,r},\, \frac{2}{75}\,g^2 - \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r,\, \frac{2}{75}\,g^2\right.\right.$$

$$\left. - \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r\right],$$

$$\left[\frac{2}{75}g^2-\frac{2}{75}g\sqrt{g^2+15r}-\frac{4}{5}r,\ -\frac{2}{5}r-\frac{14}{75}g^2+\frac{14}{75}g\sqrt{g^2+15r},\ \frac{2}{75}g^2-\frac{2}{75}g\sqrt{g^2+15r}-\frac{4}{5}r\right],$$

$$\left[\frac{2}{75}g^2-\frac{2}{75}g\sqrt{g^2+15r}-\frac{4}{5}r,\ \frac{2}{75}g^2-\frac{2}{75}g\sqrt{g^2+15r}-\frac{4}{5}r,\ -\frac{2}{5}r-\frac{14}{75}g^2+\frac{14}{75}g\sqrt{g^2+15r}\right]\Bigg]$$

$$e1:=\left[\frac{2}{15}g\sqrt{g^2+15r}-\frac{2}{15}g^2-2r,1,\{r\}\right],\left[-\frac{16}{75}g^2+\frac{16}{75}g\sqrt{g^2+15r}+\frac{2}{5}r,2,\{r,r\}\right]$$

```
> series(e1[1][1],r,2);
  series(e1[2][1],r,2);
```

$$-r+O(r^2)$$
$$2\,r+O(r^2)$$

Again, one positive eigenvalue, so "+" hexagons unstable towards stripes. Repeat for "-" hexagons:

```
> eq:=collect(subs({cos(1/2*x-sqrt(3)/2*y)=cos(-1/2*x+sqrt(3)/2*y),
  cos(1/2*x+sqrt(3)/2*y)=cos(-1/2*x-sqrt(3)/2*y)},subs(s3m,combine
  (coeff(N(a1*(cos(x)+cos(-1/2*x+sqrt(3)/2*y)+cos(-1/2*x-sqrt(3)/2*
  y))+epsilon*(alpha*cos(x)+beta*cos(-1/2*x+sqrt(3)/2*y)+gamma*cos
  (-1/2*x-sqrt(3)/2*y))),epsilon,1),trig))),{cos(x),cos(-1/2*x+sqrt
  (3)/2*y),cos(-1/2*x-sqrt(3)/2*y)}):
> A11:=factor(coeff(coeff(eq,cos(x),1),alpha,1)):
  A12:=factor(coeff(coeff(eq,cos(x),1),beta,1)):
  A13:=factor(coeff(coeff(eq,cos(x),1),gamma,1)):
  A21:=factor(coeff(coeff(eq,cos(-1/2*x+sqrt(3)/2*y),1),alpha,1)):
  A22:=factor(coeff(coeff(eq,cos(-1/2*x+sqrt(3)/2*y),1),beta,1)):
  A23:=factor(coeff(coeff(eq,cos(-1/2*x+sqrt(3)/2*y),1),gamma,1)):
  A31:=factor(coeff(coeff(eq,cos(-1/2*x-sqrt(3)/2*y),1),alpha,1)):
  A32:=factor(coeff(coeff(eq,cos(-1/2*x-sqrt(3)/2*y),1),beta,1)):
  A33:=factor(coeff(coeff(eq,cos(-1/2*x-sqrt(3)/2*y),1),gamma,1)):
  A:=matrix(3,3,[A11,A12,A13,A21,A22,A23,A31,A32,A33]);
  e1:=eigenvectors(A);
```

$$A:=\left[\left[-\frac{2}{5}r-\frac{14}{75}g^2-\frac{14}{75}g\sqrt{g^2+15r},\ \frac{2}{75}g^2+\frac{2}{75}g\sqrt{g^2+15r}-\frac{4}{5}r,\ \frac{2}{75}g^2+\frac{2}{75}g\sqrt{g^2+15r}-\frac{4}{5}r\right],\right.$$

$$\left[\frac{2}{75}g^2+\frac{2}{75}g\sqrt{g^2+15r}-\frac{4}{5}r, -\frac{2}{5}r-\frac{14}{75}g^2-\frac{14}{75}g\sqrt{g^2+15r}, \frac{2}{75}g^2+\frac{2}{75}g\sqrt{g^2+15r}-\frac{4}{5}r\right],$$

$$\left[\frac{2}{75}g^2+\frac{2}{75}g\sqrt{g^2+15r}-\frac{4}{5}r, \frac{2}{75}g^2+\frac{2}{75}g\sqrt{g^2+15r}-\frac{4}{5}r, -\frac{2}{5}r-\frac{14}{75}g^2-\frac{14}{75}g\sqrt{g^2+15r}\right]\Big]$$

$$e1 := \left[-\frac{2}{15}g\sqrt{g^2+15r}-\frac{2}{15}g^2-2r, 1, \{r\}\right], \left[-\frac{16}{75}g^2-\frac{16}{75}g\sqrt{g^2+15r}+\frac{2}{5}r, 2, \{r, r\}\right]$$

```
> series(e1[1][1],r,2);
  series(e1[2][1],r,2);
```

$$-\frac{4}{15}g^2-3r+O(r^2)$$

$$-\frac{32}{75}g^2-\frac{6}{5}r+O(r^2)$$

Both eigenvalues are negative, so "-" hexagons are stable towards stripes.

(g)

No, both stripes and "-" hexagons are stable with respect to disturbances considered.

(h)

Nonlinear equations for stripes:

```
> local gamma:
  f(x,y):=(alpha*cos(x)+beta*cos(-1/2*x+sqrt(3)/2*y)+gamma*cos
  (-1/2*x-sqrt(3)/2*y));
  eq:=collect(subs({cos(1/2*x-sqrt(3)/2*y)=cos(-1/2*x+sqrt(3)/2*y),
  cos(1/2*x+sqrt(3)/2*y)=cos(-1/2*x-sqrt(3)/2*y)},combine(N(f(x,y)
  ),trig)),{cos(x),cos(-1/2*x+sqrt(3)/2*y),cos(-1/2*x-sqrt(3)/2*y)}
  );
```

$$f(x, y) := \alpha\cos(x)+\beta\cos\left(\frac{1}{2}x-\frac{1}{2}\sqrt{3}\,y\right)+\gamma\cos\left(\frac{1}{2}x+\frac{1}{2}\sqrt{3}\,y\right)$$

$$eq := \left(-\frac{3}{2}\alpha\beta^2+\alpha r-\frac{3}{4}\alpha^3-\frac{3}{2}\alpha\gamma^2-\beta\gamma g\right)\cos(x)+\left(-\frac{3}{2}\alpha^2\beta-\frac{3}{2}\beta\gamma^2-\alpha\gamma g-\frac{3}{4}\beta^3+\beta r\right)\cos\left(\frac{1}{2}x-\frac{1}{2}\sqrt{3}\,y\right)+\left(\gamma r-\frac{3}{2}\beta^2\gamma-\alpha\beta g-\frac{3}{4}\gamma^3\right.$$

$$- \frac{3}{2}\,\alpha^2\,\gamma\Big)\cos\left(\frac{1}{2}\,x + \frac{1}{2}\,\sqrt{3}\,y\right) - \frac{1}{2}\,\alpha^2\,g - \frac{1}{2}\,\beta^2\,g - \frac{1}{2}\,\gamma^2\,g - \frac{1}{4}\,\alpha^3\cos(3\,x)$$

$$- \frac{1}{4}\,\beta^3\cos\left(\frac{3}{2}\,x - \frac{3}{2}\,\sqrt{3}\,y\right) - \frac{3}{2}\,\alpha\,\beta\,\gamma - \frac{3}{2}\,\alpha\,\beta\,\gamma\cos(\sqrt{3}\,y + x) - \frac{3}{2}\,\alpha\,\beta\,\gamma\cos($$

$$-\sqrt{3}\,y + x) - \frac{3}{2}\,\alpha\,\beta\,\gamma\cos(2\,x) - \frac{3}{4}\,\alpha^2\,\beta\cos\left(\frac{3}{2}\,x + \frac{1}{2}\,\sqrt{3}\,y\right)$$

$$- \frac{3}{4}\,\alpha^2\,\beta\cos\left(\frac{5}{2}\,x - \frac{1}{2}\,\sqrt{3}\,y\right) - \frac{3}{4}\,\alpha\,\beta^2\cos(\sqrt{3}\,y) - \frac{3}{4}\,\alpha\,\beta^2\cos(2\,x - \sqrt{3}\,y)$$

$$- \frac{3}{4}\,\alpha^2\,\gamma\cos\left(\frac{3}{2}\,x - \frac{1}{2}\,\sqrt{3}\,y\right) - \frac{3}{4}\,\alpha^2\,\gamma\cos\left(\frac{5}{2}\,x + \frac{1}{2}\,\sqrt{3}\,y\right)$$

$$- \frac{3}{4}\,\alpha\,\gamma^2\cos(\sqrt{3}\,y) - \frac{3}{4}\,\alpha\,\gamma^2\cos(2\,x + \sqrt{3}\,y) - \frac{3}{4}\,\beta^2\,\gamma\cos\left(\frac{1}{2}\,x - \frac{3}{2}\,\sqrt{3}\,y\right)$$

$$- \frac{3}{4}\,\beta^2\,\gamma\cos\left(\frac{3}{2}\,x - \frac{1}{2}\,\sqrt{3}\,y\right) - \frac{3}{4}\,\beta\,\gamma^2\cos\left(\frac{1}{2}\,x + \frac{3}{2}\,\sqrt{3}\,y\right) - \frac{3}{4}\,\beta\,\gamma^2\cos\Big(\frac{3}{2}\,x$$

$$+ \frac{1}{2}\,\sqrt{3}\,y\Big) - \cos\left(\frac{3}{2}\,x - \frac{1}{2}\,\sqrt{3}\,y\right)\alpha\,\beta\,g - \cos\left(\frac{3}{2}\,x + \frac{1}{2}\,\sqrt{3}\,y\right)\alpha\,\gamma\,g$$

$$- \cos(\sqrt{3}\,y)\,\beta\,\gamma\,g - \frac{1}{4}\,\gamma^3\cos\left(\frac{3}{2}\,x + \frac{3}{2}\,\sqrt{3}\,y\right) - \frac{1}{2}\cos(-\sqrt{3}\,y + x)\,\beta^2\,g$$

$$- \frac{1}{2}\cos(\sqrt{3}\,y + x)\,\gamma^2\,g - \frac{1}{2}\cos(2\,x)\,\alpha^2\,g$$

```
> eq1:=factor(coeff(eq,cos(x),1));
  eq2:=factor(coeff(eq,cos(-1/2*x+sqrt(3)/2*y),1));
  eq3:=factor(coeff(eq,cos(-1/2*x-sqrt(3)/2*y),1));
```

$$eq1 := -\frac{3}{2}\,\alpha\,\beta^2 + \alpha\,r - \frac{3}{4}\,\alpha^3 - \frac{3}{2}\,\alpha\,\gamma^2 - \beta\,\gamma\,g$$

$$eq2 := -\frac{3}{2}\,\alpha^2\,\beta - \frac{3}{2}\,\beta\,\gamma^2 - \alpha\,\gamma\,g - \frac{3}{4}\,\beta^3 + \beta\,r$$

$$eq3 := \gamma\,r - \frac{3}{2}\,\beta^2\,\gamma - \alpha\,\beta\,g - \frac{3}{4}\,\gamma^3 - \frac{3}{2}\,\alpha^2\,\gamma$$

(i)

The stripe solution:

```
> sl:={beta=0,gamma=0}:
  s1:=solve(subs(sl,{eq1,eq2,eq3}),alpha):
  sl:=sl union s1[2];
```

$$sl := \left\{\alpha = \frac{2}{3}\,\sqrt{3}\,\sqrt{r},\ \beta = 0,\ \gamma = 0\right\}$$

Linearization:

```
> A11:=factor(subs(sl,diff(eq1,alpha))):
```

```
A12:=factor(subs(sl,diff(eq1,beta))):
A13:=factor(subs(sl,diff(eq1,gamma))):
A21:=factor(subs(sl,diff(eq2,alpha))):
A22:=factor(subs(sl,diff(eq2,beta))):
A23:=factor(subs(sl,diff(eq2,gamma))):
A31:=factor(subs(sl,diff(eq3,alpha))):
A32:=factor(subs(sl,diff(eq3,beta))):
A33:=factor(subs(sl,diff(eq3,gamma))):
A:=matrix(3,3,[A11,A12,A13,A21,A22,A23,A31,A32,A33]);
e1:=eigenvectors(A);
```

$$A := \begin{bmatrix} -2\,r & 0 & 0 \\ 0 & -r & -\frac{2}{3}\sqrt{3}\,\sqrt{r}\,g \\ 0 & -\frac{2}{3}\sqrt{3}\,\sqrt{r}\,g & -r \end{bmatrix}$$

$$e1 := \left[\frac{1}{3}\left(-\sqrt{3}\,r + 2\,g\sqrt{r}\right)\sqrt{3},\,1,\,\{r\}\right],\ \left[-\frac{1}{3}\left(\sqrt{3}\,r + 2\,g\sqrt{r}\right)\sqrt{3},\,1,\,\{r\}\right],\ [-2\,r,\,1,\,\{r\}]$$

Now we find the stripes to be unstable towards growth of two sets of new stripes with the growth rate

$$-r + \frac{2}{3}\,g\sqrt{3}\,\sqrt{r}$$

The hexagon solution:

```
> sl:={beta=alpha,gamma=alpha}:
s1:=solve(subs(sl,{eq1,eq2,eq3}),alpha);
s3p:=subs(s1[2],sl) union s1[2];
s3m:=subs(s1[3],sl) union s1[3];
```

$$s1 := \{\alpha = 0\},\ \left\{\alpha = -\frac{2}{15}\,g + \frac{2}{15}\sqrt{g^2 + 15\,r}\right\},\ \left\{\alpha = -\frac{2}{15}\,g - \frac{2}{15}\sqrt{g^2 + 15\,r}\right\}$$

$$s3p := \left\{\alpha = -\frac{2}{15}\,g + \frac{2}{15}\sqrt{g^2 + 15\,r},\ \beta = -\frac{2}{15}\,g + \frac{2}{15}\sqrt{g^2 + 15\,r},\ \gamma = -\frac{2}{15}\,g + \frac{2}{15}\sqrt{g^2 + 15\,r}\right\}$$

$$s3m := \left\{\alpha = -\frac{2}{15}\,g - \frac{2}{15}\sqrt{g^2 + 15\,r},\ \beta = -\frac{2}{15}\,g - \frac{2}{15}\sqrt{g^2 + 15\,r},\ \gamma = -\frac{2}{15}\,g - \frac{2}{15}\sqrt{g^2 + 15\,r}\right\}$$

Linearization about the "-" hexagons:

```
> sl:=s3m;
```

```
A11:=factor(subs(sl,diff(eq1,alpha))):
A12:=factor(subs(sl,diff(eq1,beta))):
A13:=factor(subs(sl,diff(eq1,gamma))):
A21:=factor(subs(sl,diff(eq2,alpha))):
A22:=factor(subs(sl,diff(eq2,beta))):
A23:=factor(subs(sl,diff(eq2,gamma))):
A31:=factor(subs(sl,diff(eq3,alpha))):
A32:=factor(subs(sl,diff(eq3,beta))):
A33:=factor(subs(sl,diff(eq3,gamma))):
A:=matrix(3,3,[A11,A12,A13,A21,A22,A23,A31,A32,A33]);
e1:=eigenvectors(A);
```

$$sl := \left\{\alpha = -\frac{2}{15}\,g - \frac{2}{15}\sqrt{g^2+15\,r},\ \beta = -\frac{2}{15}\,g - \frac{2}{15}\sqrt{g^2+15\,r},\ \gamma = -\frac{2}{15}\,g - \frac{2}{15}\sqrt{g^2+15\,r}\right\}$$

$$A := \left[\left[-\frac{2}{5}\,r - \frac{14}{75}\,g^2 - \frac{14}{75}\,g\sqrt{g^2+15\,r},\ \frac{2}{75}\,g^2 + \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r,\ \frac{2}{75}\,g^2 + \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r\right],\right.$$
$$\left[\frac{2}{75}\,g^2 + \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r,\ -\frac{2}{5}\,r - \frac{14}{75}\,g^2 - \frac{14}{75}\,g\sqrt{g^2+15\,r},\ \frac{2}{75}\,g^2 + \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r\right],$$
$$\left.\left[\frac{2}{75}\,g^2 + \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r,\ \frac{2}{75}\,g^2 + \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r,\ -\frac{2}{5}\,r - \frac{14}{75}\,g^2 - \frac{14}{75}\,g\sqrt{g^2+15\,r}\right]\right]$$

$$e1 := \left[-\frac{2}{15}\,g\sqrt{g^2+15\,r} - \frac{2}{15}\,g^2 - 2\,r,\ 1,\ \{r\}\right],\ \left[-\frac{16}{75}\,g^2 - \frac{16}{75}\,g\sqrt{g^2+15\,r} + \frac{2}{5}\,r,\ 2,\ \{r, r\}\right]$$

```
> series(e1[1][1],r,2);
  series(e1[2][1],r,2);
```

$$-\frac{4}{15}\,g^2 - 3\,r + \mathrm{O}(r^2)$$

$$-\frac{32}{75}\,g^2 - \frac{6}{5}\,r + \mathrm{O}(r^2)$$

We find "-" hexagons to be stable. For the "+" hexagons:

```
> sl:=s3p;
```

```
A11:=factor(subs(sl,diff(eq1,alpha))):
A12:=factor(subs(sl,diff(eq1,beta))):
A13:=factor(subs(sl,diff(eq1,gamma))):
A21:=factor(subs(sl,diff(eq2,alpha))):
A22:=factor(subs(sl,diff(eq2,beta))):
A23:=factor(subs(sl,diff(eq2,gamma))):
A31:=factor(subs(sl,diff(eq3,alpha))):
A32:=factor(subs(sl,diff(eq3,beta))):
A33:=factor(subs(sl,diff(eq3,gamma))):
A:=matrix(3,3,[A11,A12,A13,A21,A22,A23,A31,A32,A33]);
e1:=eigenvectors(A);
```

$$sl := \left\{\alpha = -\frac{2}{15}\,g + \frac{2}{15}\sqrt{g^2+15\,r},\ \beta = -\frac{2}{15}\,g + \frac{2}{15}\sqrt{g^2+15\,r},\ \gamma = -\frac{2}{15}\,g + \frac{2}{15}\sqrt{g^2+15\,r}\right\}$$

$$A := \left[\left[-\frac{2}{5}\,r - \frac{14}{75}\,g^2 + \frac{14}{75}\,g\sqrt{g^2+15\,r},\ \frac{2}{75}\,g^2 - \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r,\ \frac{2}{75}\,g^2 - \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r\right],\right.$$
$$\left[\frac{2}{75}\,g^2 - \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r,\ -\frac{2}{5}\,r - \frac{14}{75}\,g^2 + \frac{14}{75}\,g\sqrt{g^2+15\,r},\ \frac{2}{75}\,g^2 - \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r\right],$$
$$\left.\left[\frac{2}{75}\,g^2 - \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r,\ \frac{2}{75}\,g^2 - \frac{2}{75}\,g\sqrt{g^2+15\,r} - \frac{4}{5}\,r,\ -\frac{2}{5}\,r - \frac{14}{75}\,g^2 + \frac{14}{75}\,g\sqrt{g^2+15\,r}\right]\right]$$

$$e1 := \left[\frac{2}{15}\,g\sqrt{g^2+15\,r} - \frac{2}{15}\,g^2 - 2\,r,\ 1,\ \{r\}\right],\ \left[-\frac{16}{75}\,g^2 + \frac{16}{75}\,g\sqrt{g^2+15\,r} + \frac{2}{5}\,r,\ 2,\ \{r, r\}\right]$$

```
> series(e1[1][1],r,2);
  series(e1[2][1],r,2);
```

$$-r + O(r^2)$$
$$2\,r + O(r^2)$$

we find a positive eigenvalue, so only the "+" haxagons are unstable.

(j)

The "-" hexagons will be selected at onset (this is a subcritical bifurcation!) because all other stripe superpositions are linearly unstable.

The new feature of this analysis is that we considered the dynamics of 3 interacting stripes rather than just a pair.

>